Problem:
Score : 400 points

Problem Statement

We have a permutation of the integers from 1 through N, p_1, p_2, .., p_N.
We also have M pairs of two integers between 1 and N (inclusive), represented as (x_1,y_1), (x_2,y_2), .., (x_M,y_M).
AtCoDeer the deer is going to perform the following operation on p as many times as desired so that the number of i (1 ≤ i ≤ N) such that p_i = i is maximized:

Choose j such that 1 ≤ j ≤ M, and swap p_{x_j} and p_{y_j}.

Find the maximum possible number of i such that p_i = i after operations.

Constraints

2 ≤ N ≤ 10^5

1 ≤ M ≤ 10^5

p is a permutation of integers from 1 through N.

1 ≤ x_j,y_j ≤ N

x_j ≠ y_j

If i ≠ j, \{x_i,y_i\} ≠ \{x_j,y_j\}.

All values in input are integers.

Input

Input is given from Standard Input in the following format:

N M
p_1 p_2 .. p_N
x_1 y_1
x_2 y_2
:
x_M y_M

Output

Print the maximum possible number of i such that p_i = i after operations.

Sample Input 1

5 2
5 3 1 4 2
1 3
5 4

Sample Output 1

2

If we perform the operation by choosing j=1, p becomes 1 3 5 4 2, which is optimal, so the answer is 2.

Sample Input 2

3 2
3 2 1
1 2
2 3

Sample Output 2

3

If we perform the operation by, for example, choosing j=1, j=2, j=1 in this order, p becomes 1 2 3, which is obviously optimal.
Note that we may choose the same j any number of times.

Sample Input 3

10 8
5 3 6 8 7 10 9 1 2 4
3 1
4 1
5 9
2 5
6 5
3 5
8 9
7 9

Sample Output 3

8

Sample Input 4

5 1
1 2 3 4 5
1 5

Sample Output 4

5

We do not have to perform the operation.

Solution in Haskell (GHC 7.10.3) (judge verdict: Accepted)

{-# OPTIONS_GHC -O2 -funbox-strict-fields #-}
import           Control.Monad               (forM, forM_, when)
import           Control.Monad.ST            (ST, runST)
import qualified Data.ByteString.Char8       as BC
import           Data.Maybe                  (fromJust)
import qualified Data.Vector.Unboxed.Mutable as VM

main :: IO ()
main = do
  [n, m] <- readInts <$> BC.getLine
  ps <- readInts <$> BC.getLine
  xys <- readIntss <$> BC.getContents
  print $ solve n m ps xys

solve :: Int -> Int -> [Int] -> [[Int]] -> Int
solve n m ps xys = runST $ do
  uf <- consUF n
  forM_ xys $ \([x, y]) ->
    unite (x - 1) (y - 1) uf
  fs <- forM (zip ps [1 .. n]) $ \(p, q) ->
    find (p - 1) (q - 1) uf
  return . length $ filter id fs

consUF :: Int -> ST s (VM.STVector s Int)
consUF n = VM.replicate n (-1)

unite :: Int -> Int -> VM.STVector s Int -> ST s Bool
unite i j uf = do
  ri <- root i uf
  rj <- root j uf
  when (ri /= rj) (updateUF ri rj uf)
  return $ ri /= rj

updateUF :: Int -> Int -> VM.STVector s Int -> ST s ()
updateUF i j uf = do
  x <- VM.read uf i
  y <- VM.read uf j
  let (i', j') = if x <= y then (i, j) else (j, i)
  VM.write uf i' $ x + y
  VM.write uf j' i'

find :: Int -> Int -> VM.STVector s Int -> ST s Bool
find i j uf = (==) <$> root i uf <*> root j uf

root :: Int -> VM.STVector s Int -> ST s Int
root i uf = do
  x <- VM.read uf i
  if x < 0
    then return i
    else do
      y <- root x uf
      VM.write uf i y
      return y

readInt :: BC.ByteString -> Int
readInt = fst . fromJust . BC.readInt

readInts :: BC.ByteString -> [Int]
readInts = map readInt . BC.words

readIntss :: BC.ByteString -> [[Int]]
readIntss = map readInts . BC.lines